# Maintainer: Antonio Rojas <arojas@archlinux.org>

_pipname=database_cubic_hecke
pkgname=python-${_pipname//_/-}
pkgver=2022.4.4
pkgrel=2
pkgdesc='Representations of the cubic Hecke algebra on 4 strands as Python dictionaries'
arch=(any)
url='https://github.com/soehms/database_cubic_hecke'
license=(GPL)
depends=(python)
makedepends=(python-build python-installer python-setuptools-scm python-wheel)
source=(https://pypi.io/packages/source/d/$_pipname/$_pipname-$pkgver.tar.gz)
sha256sums=('553654a4ce987a277fe956a9a450d738bd1f58b96c45499075e28f2bca927ae9')

build() {
  cd $_pipname-$pkgver
  python -m build --wheel --no-isolation
}

package() {
  cd $_pipname-$pkgver
  python -m installer --destdir="$pkgdir" dist/*.whl
}
